
Abstract. A discriminator partitions values associated with keys into
groups listed in ascending order. Discriminators can be defined generi-
cally by structural recursion on representations of ordering relations. Em-
ploying type-indexed families we demonstrate how tries with an optimal-
time lookup function can be constructed generically in worst-case linear
time. We provide generic implementations of comparison, sorting, dis-
crimination and trie building functions and give equational proofs of
correctness that highlight core relations between these algorithms.

1 Introduction

Sorting and searching are some of the most fundamental topics in computer
science. In this paper we define generic functions for solving sorting and searching
problems, based on distributive, that is “radix-sort-like”, techniques rather than
comparison-based techniques. The functions are indexed by representations of
ordering relations on keys of type K. In each case the input is an association
list of key-value pairs, and the values are treated as satellite data, that is, the
functions are parametric in the value type V. Intuitively, this means values are
pointers that are not dereferenced during execution of these functions [1]. We
identify a hierarchy of operations:1

sort :: Order k→ [k× v]→ [v]
discr :: Order k→ [k× v]→ [ [v] ]
trie :: Order k→ [k× v]→ Trie k [v]

The sorting function, sort, outputs the value components according to the given
order on K without, however, returning the key component. For example,

� sort (OList OChar) [("ab", 1), ("ba", 2), ("abc", 3), ("ba", 4)]
[1, 3, 2, 4] ,

1 Executable code is rendered in Haskell, which requires lower-case identifiers for type
variables. We use the corresponding upper-case identifiers in the running text and
in program calculations.

where OList OChar denotes the standard lexicographic order on strings. We re-
quire that sort be stable in the sense that the relative order of values with equiv-
alent keys is preserved. Discarding the keys may seem surprising and restrictive
at first. Nothing is lost, however, since parametricity allows us to arrange it so
that the keys are also returned. We simply associate the keys with themselves.

� sort (OList OChar) [("ab", "ab"), ("ba", "ba"), ("abc", "abc"), ("ba", "ba")]
["ab", "abc", "ba", "ba" ]

The discriminator, discr, outputs the value components grouped into runs of
values associated with equivalent keys. For example,

� discr (OList OChar) [("ab", 1), ("ba", 2), ("abc", 3), ("ba", 4)]
[[1], [3], [2, 4]] .

The trie constructor, trie, outputs a trie that can subsequently be efficiently
searched for values associated to a particular key. The type of trie constructed
depends on the type of the keys. For example,

� let t = trie (OList OChar) [("ab", 1), ("ba", 2), ("abc", 3), ("ba", 4)]
� lookup t "ba"
Just [2, 4] .

The function discr was introduced by Henglein [2,3] (originally called sdisc).
It provides a framework for bootstrapping any base sorting algorithm for a finite
type, such as bucket sort, to a large class of user-definable orders on first-order
and recursive types. To this end it employs a strategy corresponding to most-
significant-digit (MSD) in radix sorting.

The functions sort and trie are novel. Algorithmically, sort does the same
as discr, but employing a least-significant-digit (LSD) strategy. Drawing on the
informal correspondence of MSD radix sort with tries [4, p. 3], trie generalizes
discr and generates the generalized tries introduced by Hinze [5]. It subsumes
discr (which in turn subsumes sort) in the sense that it executes in the same
time (usually linear in the size of the input keys), but additionally facilitates
efficient search for values associated with any key.

In this paper we make the following novel contributions:

– We show that a function of type [K×V ]→ [V ] is a stable sorting function
if and only if it is strongly natural in V, preserves singleton lists, and sorts
lists of length 2 correctly. A function is strongly natural if it commutes with
filtering, that is, the removal of elements from a list.

– We give new generic definitions of: sort, which generalizes least-significant-
digit (LSD) radix sort to arbitrary types and orders definable by an expres-
sive language of order representations ; and trie, which generalizes discr to
construct efficiently key-searchable tries. Both run in worst-case linear time
for a large class of orders.

– We provide equational proofs for sort o being a stable sorting function and
show that sort o = concat · discr o and discr o = flatten · trie o for all

inductively defined order representations o, where concat is list concatenation
and flatten lists the values stored in a trie in ascending key order. The first
equality is nontrivial as discr and sort have different underlying algorithmic
strategies for product types: MSD versus LSD. The proof highlights the
strong naturality properties of sort and discr.

– We offer preliminary benchmark results of our generic distributive sort-
ing functions, which are surprisingly promising when compared to Haskell’s
built-in comparison-based sorting function.

The paper focuses on and highlights the core relations between these algo-
rithms, notably the role of strong naturality. Here we limit ourselves to a re-
stricted class of orders and leave asymptotic analysis, performance engineering,
and a proper empirical performance analysis for future work. But certainly some
benchmarks are not amiss to whet the appetite. The task is to sort the words of
Project Gutenberg’s The Bible, King James Version (5218802 characters, 824337
words). We compare Haskell’s built-in sortBy called with Haskell’s own compare
and our generically defined comparison function cmp o, to generic sorting and
generic discrimination, and to sorting via generic tries.

0 1 2 3 4 5

sortBy compare
sortBy (cmp o)

sort o
concat · discr o

concat · flatten · trie o

4.01

5.1

2.34

1.16

1.68

We assume familiarity with the programming language GHC Haskell and basic
notions of category theory. Unless noted otherwise, we work in Set, the category
of sets and total functions.

2 Order Representations

Comparison-based sorting and searching methods are attractive because they
easily generalize to arbitrary orders: simply parameterize the program code for,
say, Quicksort [6] over its comparison function, and apply it to a user-defined
ordering leq :: T → T → �. An analogous approach works for searching on T
using, say, red-black trees [7,8]. While maximally expressive, specifying orders
via such “black-box” binary comparisons, has two disadvantages:

1. Deliberately or erroneously, leq may not implement a total preorder.
2. Both sorting and searching are subject to lower bounds on their performance:

sorting requires Ω(n log n) comparisons, and searching for a key requires
Ω(log n), where n is the number of keys in the input.

However, theoretically and practically faster distributive methods are known for
certain orders, notably radix sorting and tries.

As shown by Henglein [3], many orders can be denoted by order representa-
tions, constructors for building new orders from old:

data Order :: ∗ → ∗ where
OUnit :: Order ()
OSum :: Order k1 → Order k2 → Order (k1 + k2)
OProd :: Order k1 → Order k2 → Order (k1 × k2)
OMap :: (k1 → k2)→ (Order k2 → Order k1)
OChar :: Order Char -- 7 bit ASCII .

Here OUnit denotes the trivial order on the unit type. OSum o1 o2 represents the
lexicographic order on tagged values such that Inl-tagged values are less than Inr-
tagged values, and values with the same tag are ordered by o1 or o2, depending
on the tag. OProd o1 o2 denotes the lexicographic order on pairs, ordering pairs
according to their first component, where pairs with equivalent first component
are ordered according to their second component. OMap f o orders the domain
of f according to the order o on its codomain. Note that OMap is contravariant.
Finally, OChar denotes the standard order on 7-bit ASCII characters.

The OMap-constructor adds considerable expressiveness. For example,

rprod :: Order k1 → Order k2 → Order (k1 × k2)
rprod o1 o2 = OMap (λ(a, b)→ (b, a)) (OProd o2 o1)

specifies the lexicographic order on pairs based on the second component as the
dominant one. Similarly, rsum can be specified, which orders Inr-tagged values
as less than Inl-tagged ones.

Order representations are terms that can be treated inductively as finite trees
or coinductively as potentially infinite trees.

The coinductive approach permits definition of orders for recursive data types
by guarded recursion. For example,

olist :: Order k→ Order [k]
olist o = os where os = OMap out (OSum OUnit (OProd o os))
out :: [a]→ () + a× [a]
out [ ] = Inl ()
out (a : as) = Inr (a, as)

defines the standard lexicographic order on lists, based on the element order o.
Henglein [3] takes the coinductive approach with additional order constructors
for inverse, multiset and set orders.

In the inductive approach we can add new constructors explicitly:

OList :: Order k→ Order [k]

or, more generally, employ an explicit fixed point operator [2].
The expressiveness of order representations is orthogonal to the aims of this

paper. For simplicity we assume the inductive approach and concentrate on sums
and products.

3 Generic Comparison

We require order representations to denote total preorders.

leq :: Order k→ (k→ k→ �)
leq OUnit a b = True
leq (OSum o1 o2) a b = case (a, b) of

(Inl a1, Inl a2 )→ leq o1 a1 a2
(Inl , Inr )→ True
(Inr , Inl )→ False
(Inr b1, Inr b2)→ leq o2 b1 b2

leq (OProd o1 o2) a b = leq o1 (fst a) (fst b) ∧
(leq o1 (fst b) (fst a) =⇒ leq o2 (snd a) (snd b))

leq (OMap g o) a b = leq o (g a) (g b)
leq (OChar) a b = a � b

leq o is indeed a total preorder; it is transitive and total, leq o x y ∨ leq o y x.
Because of totality the case for OProd can also be written as

leq (OProd o1 o2) a b = if leq o2 (snd a) (snd b) then leq o1 (fst a) (fst b)
else ¬ (leq o1 (fst b) (fst a))

This variant is strict (leq o1 and leq o2 are called), but it calls leq o1 only once.
The first variant is lazy (leq o2 is not necessarily called), but possibly calls leq o1
twice.

The function leq implements a two-way comparison; a more useful function is
cmp::Order k→ (k→ k→ Ordering), which implements a three-way comparison
and avoids the double traversal in the product case.

4 Generic Distributive Sorting

Generic sorting takes a list of key-value pairs and returns the values in non-
decreasing order of their associated keys. The keys are discarded in the course
of this process. The idea is that, barring OMap in order representations, each
component of each key is touched exactly once. Consequently, the running time
of sort is proportional to the total size of the keys (again, ignoring OMap).

sort :: Order k→ [k× v]→ [v]
sort o [ ] = [ ]
sort (OUnit) rel = map val rel
sort (OSum o1 o2) rel = sort o1 (filter froml rel) ++ sort o2 (filter fromr rel)
sort (OProd o1 o2) rel = sort o1 (sort o2 (map curryr rel))
sort (OMap g o) rel = sort o (map (g× id) rel)
sort (OChar) rel = bucketSort (’\NUL’, ’\DEL’) rel

Like generic comparison, sort is indexed by order representations. It is further-
more parametric in the type of values. Let us discuss each case in turn.

The first equation is vital for the coinductive approach to recursive types. It
is necessary to ensure that for each recursive invocation of sort the total size of
the keys is strictly decreasing.

For the unit type there is little to do: we simply discard the keys using val
defined val (k, v) = v.

The case for sums takes an approach à la Quicksort: the input list is parti-
tioned into a list whose keys are of the form Inl k1 and a second list whose keys
are of the form Inr k2. The constructors are discarded, the sub-lists are sorted
recursively using the appropriate orders, and the final results are concatenated.
(As an aside, in the partitioning phase we touch the keys actually twice, but this
is easily avoided by combining the two sweeps into a single one.) The function
filter::(a→ Maybe b)→ ([a]→ [b]), called mapMaybe elsewhere, combines map-
ping and filtering: if the argument function returns a value of the form Just b,
then b is included in the output list. If the result is Nothing, well, nothing is
added. (Don’t confuse our filter with Haskell’s filter :: (a→ �)→ ([a ]→ [a ])).

filter :: (a→ Maybe b)→ ([a]→ [b])
filter p xs = [y | x← xs, Just y← [p x ] ]

The function froml ::(k1+k2×v)→ Maybe (k1×v) maps (Inl k1, v) to Just (k1, v),
and (Inr k2, v) to Nothing. The function fromr is defined analogously.

The most interesting case is the one for products. The natural isomorphism
curryr : (K1 × K2) × V ∼= K2 × (K1 × V) shifts the more significant part of
the key into the value component. Then sort is called twice: the first invocation
sorts according to o2 discarding the K2 part, the second sorts according to o1
discarding the K1 component. For this to be correct, sort o1 had better be stable;
we shall return to this point below. Furthermore, sort relies on polymorphic
recursion: the first call to sort instantiates V to K1 ×V.

For OMap we simply apply the key transformation using map (g × id) and
then sort the transformed keys-value pairs.

Characters are sorted using bucket sort, which can be seen as a specialization
of sort (actually of discr introduced in Section 5) for enumeration types.

bucketSort :: (Bounded i, Ix i)⇒ (i, i)→ [(i, v)]→ [v]
bucketSort bs rel = concat (elems (accumArray (λws w→ ws ++ [w]) [ ] bs rel))

(Here ++ is used for clarity; in our implementation it is replaced by a constant-
time operation.) Any other algorithm for sorting characters or, for that matter,
other primitive types could be used. The particular algorithm invoked can even
be made data dependent ; for small rel we might choose insertion sort instead
of bucket sort to avoid sparse bucket table traversals. The key point is that
sort reduces a sorting problem to basic sorting on finite domains. Conversely, it
extends distributive sorting from their restricted domains such as small integers
and character strings to arbitrary orders definable by order representations.

Let us now turn to the correctness of sort. Its implementation builds upon
standard components, except perhaps the case for sums, which relies on the func-
tion filter. Note that filter takes a partial function as an argument, represented

by a total function of type A → Maybe B, an arrow in the Kleisli category in-
duced by the monad Maybe. For the proofs it will be convenient to actually work
in this Kleisli category (but only for the arguments of filter). In the calculations,
we signal these steps by the hint “Kleisli:”. A few remarks are in order.

Working in the Kleisli category has the advantage that the notation is fairly
light-weight: we write id rather than η, and we write p · q for the Kleisli com-
position of p and q rather than μ · Maybe p · q. We also silently embed total
functions into the Kleisli category: filter f really means filter (η · f). In any case,
there is little room for confusion since we let f, g . . . range over total functions
and p, q over partial functions. The product × can be lifted to a binary functor ⊗
over the Kleisli category, which is, however, not a categorical product. Rather,
⊗ is a so-called tensor product, a binary functor which is coherently associative
and commutative. We overload × to denote both the product in the underlying
category and the tensor product.

A partial function that we will use time and again is inl◦, the left-inverse of
inl. The partial function froml can be neatly expressed in terms of inl◦: we have
froml = inl◦ ⊗ id, or just inl◦ × id. Likewise, fromr = inr◦ ⊗ id.

The function filter satisfies a variety of properties. First and foremost, it is a
monoid homomorphism:

filter p [ ] = [ ] , (4.1a)
filter p (xs ++ ys) = filter p xs ++ filter p ys . (4.1b)

Furthermore, filter is functorial, taking Kleisli arrows to arrows in the underlying
category. Formally, filter is the arrow part of the functor Filter :Kleisli→ Set,
whose object part is defined Filter A = List A. In other words, filter preserves
identity and composition.

filter id = id (4.2a)
filter (p · q) = filter p · filter q (4.2b)

Moreover, if its first argument is a total function, then filter f is just List f.
Most of the following proofs will be conducted in a point-free style. For refer-

ence, here is a suitably reworked version of sort.

sort (OUnit) = List val
sort (OSum o1 o2) = sort o1 · filter (inl◦ × id) ++ sort o2 · filter (inr◦ × id)
sort (OProd o1 o2) = sort o1 · sort o2 · List curryr

The sum case uses a lifted variant of append, (f ++ g) x = f x ++ g x, overloading
the operator ++ to denote both the lifted and the unlifted version.

4.1 Naturality

A vital property of sort o is that it is natural in the type of values, sort o : List ◦
(K× ) →̇ List, that is

List f · sort o = sort o · List (id× f) , (4.3)

for all f : A→ B.

Because of naturality it is sufficient to show that the instance sort o:[K×� ]→
[� ] works correctly. (Recall that values can be seen as pointers; natural numbers
are like unique pointers.) Formally, sort o is fully determined by this instance:

sort o [(k1, v1), . . . , (kn, vn)]
= { let ix : �→ V be an indexing function so that ix i = ki }

sort o (List (id× ix) [(k1, 1), . . . , (kn, n)])
= { sort is natural (4.3) }

List ix (sort o [(k1, 1), . . . , (kn, n)]) .

In the last equation sort o is used at instance �.
Of course, the statement that sort o is natural requires proof. Actually, sort

satisfies a much stronger property, which we discuss next.

4.2 Strong Naturality

Property (4.3) remains valid if we replace List by filter:

filter p · sort o = sort o · filter (id× p) , (4.4)

for all p : A→ Maybe B. It does not matter whether we filter before or after an
invocation of sort o, as long as filter only refers to the values, and not the keys.

Now, since filter is the arrow part of a functor between the Kleisli category of
Maybe and the underlying category, (4.4) also amounts to a naturality property,
sort o : Filter ◦ (K ⊗ ) →̇ Filter. A simple consequence of what we call strong
naturality (4.4) is that sort preserves the empty list.

Turning to the proof of (4.4), we proceed by induction over the structure of
order representations.
Case o = OUnit:

filter p · sort OUnit
= { definition of sort }

filter p · List val
= { Kleisli: p · val = val · (id × p) }

List val · filter (id × p)
= { definition of sort }

sort OUnit · filter (id × p)

Recall that × aka ⊗ is not a categorical product in the Kleisli category, we have
val · (id× p) = p · val, but not key · (id × p) = id · key where key (k, v) = k.
Case o = OSum o1 o2: the central step is the fourth one, where we swap two
filters, one that acts on the keys and a second that acts on the values.

filter p · sort (OSum o1 o2)
= { definition of sort }

filter p · (sort o1 · filter (inl◦ × id) ++ . . .)
= { filter is a monoid homomorphism (4.1b) }

filter p · sort o1 · filter (inl◦ × id) ++ . . .
= { ex hypothesi }

sort o1 · filter ((id + id)× p) · filter (inl◦ × id) ++ . . .
= { Kleisli: × is a binary functor }

sort o1 · filter (inl◦ × id) · filter ((id + id)× p) ++ . . .
= { fusion: f · h ++ g · h = (f ++ g) · h }

(sort o1 · filter (inl◦ × id) ++ . . .) · filter ((id + id)× p)
= { definition of sort }

sort (OSum o1 o2) · filter ((id + id)× p)

Note that the lifting of + is a categorical coproduct in the Kleisli category.
Case o = OProd o1 o2: note that Property (4.4) is universally quantified over
all p. This is essential as sort relies on polymorphic recursion: the second use of
the induction hypothesis (∗) instantiates p to id× p.

filter p · sort (OProd o1 o2)
= { definition of sort }

filter p · sort o1 · sort o2 · List curryr
= { ex hypothesi }

sort o1 · filter (id× p) · sort o2 · List curryr
= { ex hypothesi (∗) }

sort o1 · sort o2 · filter (id × (id × p)) · List curryr
= { Kleisli: (q× (p× r)) · curryr = curryr · ((p× q)× r) }

sort o1 · sort o2 · List curryr · filter ((id × id)× p)
= { definition of sort }

sort (OProd o1 o2) · filter ((id × id)× p)

The natural isomorphisms val:1×K ∼= K and curryr:(K1×K2)×V ∼= K2×(K1×V)
are also natural isomorphisms in the Kleisli category.

4.3 Correctness: Permutation

Our first goal is to show that sort o produces a permutation of the input values.
Perhaps surprisingly, it suffices to show that sort o permutes one-element lists!
We already know that it is sufficient to show the correctness of a particular
instance: sort o [(k1, 1), . . . , (kn, n)]. Now, let �i� be the partial function that
maps i to i and is undefined otherwise. Let 1 � i � n, then

filter �i� (sort o [(k1, 1), . . . , (kn, n)])
= { sort is strongly natural (4.4) }

sort o (filter (id× �i�) [(k1, 1), . . . , (kn, n)])
= { filter (id × �i�) [(k1, 1), . . . , (kn, n)] = [(ki, i)] }

sort o [(ki, i)]
= { proof obligation: sort permutes 1-element lists (4.5) }

[i] .

Thus, sort o outputs each index exactly once; in other words, it permutes the
input list. The proof obligation (recall that return a = [a])

sort o · return = return · val (4.5)

is easy to discharge and left as an instructive exercise to the reader.

4.4 Correctness: Ordered

Our second task is to show that the values are output in non-decreasing order
of their associated keys. We aim to show

sort o [ . . . , (ki, i), . . . , (kj, j), . . . ] = [ . . . , i, . . . , j, . . . ] ⇐⇒ leq o ki kj .

Due to strong naturality, it suffices to show that sort o works correctly on two-
element lists! Let �i, j� be the partial function that maps i to i and j to j and is
undefined otherwise. Let 1 � i, j � n, then

filter �i, j� (sort o [(k1, 1), . . . , (kn, n)]) = [i, j]
⇐⇒ { sort is strongly natural (4.4) }

sort o (filter (id × �i, j�) [(k1, 1), . . . , (kn, n)]) = [i, j]
⇐⇒ { filter (id × �i, j�) [(k1, 1), . . . , (kn, n)] = [(ki, i), (kj, j)] }

sort o [(ki, i), (kj, j)] = [i, j ]
⇐⇒ { proof obligation: sort sorts 2-element lists (4.6) }

leq o ki kj .

Thus, sort o outputs i before j if and only if leq o ki kj. Since we already know
that sort o permutes its input, this implies the correctness of sort o.

It remains to show that sort treats 2-element lists correctly: let i  j, then

sort o [(a, i), (b, j)] = [i, j] ⇐⇒ leq o a b . (4.6)

Since only a small finite number of cases have to be considered, this is a simple
exercise, which we relegate to Appendix A.

5 Generic Discrimination

A discriminator returns a list of non-empty lists of values, where the inner lists
group values whose keys are equivalent. Again, the keys are discarded in the
process, but, barring OMap in order representations, this time each component
of each key is touched at most once.

discr :: Order k → [k × v ] → [[v ]]
discr o [ ] = [ ]
discr o [(k, v)] = [[v ]]
discr OUnit rel = [map val rel ]
discr (OSum o1 o2) rel = discr o1 (filter froml rel) ++ discr o2 (filter fromr rel)
discr (OProd o1 o2) rel = concat (map (discr o2) (discr o1 (map curryl rel)))
discr (OMap g o) rel = discr o (map (g × id) rel)
discr (OChar) rel = bucketDiscr (’\NUL’, ’\DEL’) rel

In the unit case we have identified a group of key-value pairs whose keys are
equivalent, even identical. This group is returned as a singleton list. The sum case
is the same as for sort. The most interesting case is again the one for products.

The natural isomorphism curryl : (K1 × K2) × V ∼= K1 × (K2 × V) also known
as assoc shifts the least significant part of the key into the value component.
The resulting list is discriminated according to o1, each of the resulting groups
is discriminated according to o2, and finally the nested groups are flattened. The
definition of discr has an additional base case for the singleton list. This may
improve the performance dramatically since the key component in the argument
need not be traversed. For lexicographic string sorting this specializes to the
property of MSD radix sort, which only traverses the minimum distinguishing
prefixes of the strings, which may be substantially fewer characters than their
total number. Finally, characters are sorted using bucket sort—this time we
simply return the list of non-empty buckets.

bucketDiscr :: (Bounded i, Ix i)⇒ (i, i)→ [i× v ]→ [ [v] ]
bucketDiscr bs rel

= [xs | xs← elems (accumArray (λws w→ ws ++ [w]) [ ] bs rel),¬ (null xs)]

As for sort, any other base type discriminator could be plugged in.

5.1 Correctness

If we concatenate the groups returned by a generic discriminator, we obtain
generic sorting.

concat · discr o = sort o (5.1)

The proof is straightforward for units and sums, the interesting case is again the
one for products. For products discrimination works from left to right, whereas
sorting proceeds right to left. This means we have to be able to swap operations:

concat · discr (OProd o1 o2)
= { definition of discr }

concat · concat · List (discr o2) · discr o1 · List curryl
= { monad law }

concat · List concat · List (discr o2) · discr o1 · List curryl
= { ex hypothesi }

concat · List (sort o2) · discr o1 · List curryl
= { proof obligation: see below }

concat · discr o1 · sort o2 · List curryr
= { ex hypothesi }

sort o1 · sort o2 · List curryr
= { definition of sort }

sort (OProd o1 o2) .

The property used in the central step, being able to swap sort o2 and discr o1,
is actually not specific to sort. Our generic discriminators commute with every
strong natural transformation. Let discr o:[K×V]→ [ [V ] ] and π:[A×V]→ [V ].
If π is strongly natural, filter p · π = π · filter (id× p), then

List π · discr o = discr o · π · List swap , (5.2)

where swap : K1 × (K2 ×V) ∼= K2 × (K1 ×V). Note that swap · curryl = curryr.
Case o = OUnit:

List π · discr OUnit : [1× (A×V)]→ [ [V ] ]
= { definition of discr }

List π · return · List val
= { return is natural: List f · return = return · f }

return · π · List val
= { Kleisli: val = (id × val) · swap }

return · π · List (id × val) · List swap
= { assumption: π is (strongly) natural }

return · List val · π · List swap
= { definition of discr }

discr OUnit · π · List swap

Case o = OSum o1 o2:

List π · discr (OSum o1 o2) : [(K1 + K2)× (A×V)]→ [ [V ] ]
= { definition of discr }

List π · (discr o1 · filter (inl◦ × id) ++ . . .)
= { List π is a monoid homomorphism }

List π · discr o1 · filter (inl◦ × id) ++ . . .
= { ex hypothesi }

discr o1 · π · List swap · filter (inl◦ × id) ++ . . .
= { Kleisli: swap · (p× (q× r)) = (q× (p× r)) · swap }

discr o1 · π · filter (id × (inl◦ × id)) · List swap ++ . . .
= { assumption: π is strongly natural }

discr o1 · filter (inl◦ × id) · π · List swap ++ . . .
= { fusion: f · h ++ g · h = (f ++ g) · h }

(discr o1 · filter (inl◦ × id) ++ . . .) · π · List swap
= { definition of discr }

discr (OSum o1 o2) · π · List swap

Case o = OProd o1 o2:

List π · discr (OProd o1 o2) : [(K1 ×K2)× (A×V)]→ [ [V ] ]
= { definition of discr }

List π · concat · List (discr o2) · discr o1 · List curryl
= { concat is natural: List f · concat = concat · List (List f) }

concat · List (List π · discr o2) · discr o1 · List curryl
= { ex hypothesi }

concat · List (discr o2 · π · List swap) · discr o1 · List curryl
= { discr o is natural: List (List f) · discr o = discr o1 · List (id × f) }

concat · List (discr o2 · π) · discr o1 · List ((id × swap) · curryl)
= { ex hypothesi }

concat · List (discr o2) · discr o1 · π · List (swap · (id × swap) · curryl)
= { Kleisli: swap · (id × swap) · curryl = (id × curryl) · swap }

concat · List (discr o2) · discr o1 · π · List ((id × curryl) · swap)
= { assumption: π is (strongly) natural }

concat · List (discr o2) · discr o1 · List curryl · π · List swap
= { definition of discr }

discr (OProd o1 o2) · π · List swap

Of course, discr o itself is also strongly natural:

List (filter p) · discr o = discr o · filter (id × p) , (5.3)

for all p : A→ Maybe B. The proof is similar to the one for sort.

6 Generic Distributive Searching

Let us now turn to distributive searching using tries. In this paper we concentrate
on bulk operations such as trie and flatten. One-at-a-time operations such as
lookup and insert have been described elsewhere [5]. It turns out that trie is
very similar to discr—we essentially replace ++ and concat by trie constructors.
This move retains more of the original information, which is vital for supporting
subsequent efficient random access to the values associated with a key. By storing
the keys together with the values in the trie, we can even recreate all of the
original keys. (If OMap’s key transformations are injective this can even be done
without explicitly storing the keys.)

For every order representation there is a corresponding trie constructor. Ad-
ditionally, we have an empty trie, which is important for efficiency reasons [5].

data Trie :: ∗ → ∗ → ∗ where
TEmpty :: Trie k v
TUnit :: v→ Trie () v
TSum :: Trie k1 v→ Trie k2 v→ Trie (k1 + k2) v
TProd :: Trie k1 (Trie k2 v)→ Trie (k1, k2) v
TMap :: (k1 → k2)→ (Trie k2 v→ Trie k1 v)
TChar :: Char.Trie v→ Trie Char v

A trie of type Trie K V represents a finite mapping from K to V, sometimes
written VK. The cases for unit, sums, and products are based on the law of
exponentials: V1 ∼= V, VK1+K2 ∼= VK1 × VK2 , and VK1×K2 ∼= (VK2)K1 . The
second but last case is interesting: the counterpart of OMap is TMap, which
retains the key transformation. This is necessary when searching for a key that is
subject to an OMap-order. Finally, we assume the existence of a suitable library,
Char, implementing finite maps with character keys; for instance, character-
indexed arrays, simple lists, binary trees (the basis of ternary tries). Indeed,
depending on the actual data encountered, multiple data structures may even
be mixed.

A trie for a given key type is a functor.

instance Functor (Trie k) where
fmap f (TEmpty) = TEmpty
fmap f (TUnit v) = TUnit (f v)

fmap f (TSum t1 t2) = TSum (fmap f t1) (fmap f t2)
fmap f (TProd t) = TProd (fmap (fmap f) t)
fmap f (TMap g t) = TMap g (fmap f t)
fmap f (TChar t) = TChar (fmap f t)

The operation flatten lists the values stored in a trie.

flatten :: Trie k v→ [v]
flatten (TEmpty) = [ ]
flatten (TUnit v) = [v]
flatten (TSum t1 t2) = flatten t1 ++ flatten t2
flatten (TProd t) = concatMap flatten (flatten t)
flatten (TMap g t) = flatten t
flatten (TChar t) = Char.flatten t

It is natural in V, that is, flatten : Trie K →̇ List.
The operation trie turns a finite relation, represented by an association list

of type [K × V ], into a finite list-valued map, represented by a trie of type
Trie K [V ].

trie :: Order k→ [k× v]→ Trie k [v]
trie o [ ] = TEmpty
trie OUnit rel = TUnit (map val rel)
trie (OSum o1 o2) rel = TSum (trie o1 (filter froml rel)) (trie o2 (filter fromr rel))
trie (OProd o1 o2) rel = TProd (fmap (trie o2) (trie o1 (map curryl rel)))
trie (OMap g o) rel = TMap g (trie o (map (g× id) rel))
trie (OChar) rel = TChar (Char.trie rel)

As we have noted before, trie arises out of discr by replacing ++, concat etc by
the appropriate trie constructors.

Indeed, if we ‘undo’ the transformation using flatten, we obtain the generic
discriminator.

discr o = flatten · trie o (6.1)

The straightforward proof can be found in Appendix B.

7 Related Work

Drawing on the algorithmic techniques termed multiset discrimination developed
by Paige and others [9], Henglein [2,3] has shown how to make MSD distributive
sorting generic by introducing generic discriminators, which have linear-time
performance over a rich class of orders.

Building on the work of Connelly and Morris [10], Hinze [5] pioneered the
type-indexed tries we produce. Here they are extended to support orders defined
with OMap. The generic LSD distributive sorting and trie building functions
developed here are also new. In particular, trie constructs a trie in bulk without
incurring the substantial update costs of one-by-one insertion.

Wadler [11] derives as a “free theorem” that any function parametric in bi-
nary comparison � commutes with map f if the function f is an order embedding,
f x �′ f y ⇐⇒ x � y; this includes all comparison-based sorting algorithms. As
shown by Day et al. [12], Knuth’s 0-1 principle for sorting networks [13] can
also be seen as a free theorem for sorting networks formulated as comparator-
parametrized functions. These properties correspond to naturality, respectively
parametricity properties on the keys; they are different from our naturality prop-
erty 4.3 since the latter applies to the value components and leaves the key
components invariant.

Our strong naturality property 4.4, coupled with preserving singletons and
correct sorting of two-element lists, corresponds to Henglein’s consistent permu-
tativity, which characterizes stable sorting functions [14]. It is, however, a more
general and a more elegant formulation supporting equational reasoning. In par-
ticular, it highlights the semantic benefits of adopting a formulation for sorting
based on key-value pairs rather than keys alone.

Gibbons [15] shows how an LSD radix sort for lists can be derived from a stable
MSD radix sort that first builds an explicit trie and then flattens it into the result
list. Since MSD radix sort, even with explicit tries, is sometimes preferable to
LSD radix sort (to avoid sparse bucket table traversal [16] and for large data
sets [17]), the derivation makes sense in both directions. Our development can
be seen as a generalization of Gibbons’ work: it works for arbitrary denotable
orders over any type; we decompose MSD sorting into discrimination followed
by concatenation, without the need for a trie (though it can be achieved by
way of trie); and our commutativity property 5.2 holds for any strong natural
transformation, not just for sorting functions.

8 Conclusion

Comparison-based sorting algorithms and search trees are easily made generic,
that is, applicable to user-defined orders, by abstraction over the comparison
function. This has arguably contributed to their popularity even though dis-
tributive (radix/trie) and hashing techniques often have superior performance
for special types, such as machine integers and character strings.

We have shown how to construct generic comparison, sorting, discrimination
and trie building operations by induction over a class of orders including standard
orders on primitive types; lexicographic orders on sums, products and lists; and
orders defined as the inverse image of a given order under an arbitrary function.

We have identified strong naturality—commutativity with filtering—as a pow-
erful property of stable sorting functions, and shown discrimination to commute
with any strongly natural transformation, including, but not limited to, stable
sorting functions.

The trie building operation yields a data structure that is not only asymptot-
ically as efficient as discrimination but also supports efficient key-based random
access, without incurring a one-at-a-time insertion overhead during construction.

Future work consists of extending equational reasoning and calculational cor-
rectness proofs to coinductive order representations; investigating data-dependent
variations of our generic functions and staged execution for our data-independent
generic functions by compile-time specialization (partial evaluation) and ex-
ploiting parallelism at word, multicore, and manycore/GPU levels; and eventu-
ally providing architecture-independent frameworks encapsulating distributive
sorting and searching methods as semantically (obeying representation inde-
pendence) and computationally (exhibiting superior performance) well-behaved
alternatives to comparison-based methods.

A Proof of Property (4.6)

The cases for unit and sums are straightforward. We only consider the product
case, which is actually instructive.
Case o = OProd o1 o2: again, we make use of naturality to be able to apply the
induction assumption.

sort (OProd o1 o2) [((a1, a2), i), ((b1, b2), j)] = [i, j]
⇐⇒ { definition of sort }

sort o1 (sort o2 (List curryr [((a1, a2), i), ((b1, b2), j)])) = [i, j]
⇐⇒ { definition of curryr }

sort o1 (sort o2 [(a2, (a1, i)), (b2, (b1, j))]) = [i, j]
⇐⇒ { let re i = (a1, i) and re j = (b1, j) }

sort o1 (sort o2 (List (id × re) [(a2, i), (b2, j)]) = [i, j]
⇐⇒ { sort is natural (4.3) }

sort o1 (List re (sort o2 [(a2, i), (b2, j)]) = [i, j]

The strict version of leq suggests to conduct a case analysis on leq o2 a2 b2.

Case leq o2 a2 b2:

⇐⇒ { ex hypothesi }
sort o1 (List re [i, j ]) = [i, j ]

⇐⇒ { definition of re }
sort o1 [(a1, i), (b1, j)] = [i, j ]

⇐⇒ { ex hypothesi }
leq o1 a1 b1

⇐⇒ { definition of leq }
leq (OProd o1 o2) (a1, a2) (b1, b2)

Case ¬ (leq o2 a2 b2):

⇐⇒ { ex hypothesi }
sort o1 (List re [j, i ]) = [i, j ]

⇐⇒ { definition of re }
sort o1 [(b1, j), (a1, i)] = [i, j ]

⇐⇒ { ex hypothesi }
¬ (leq o1 b1 a1)

⇐⇒ { definition of leq }
leq (OProd o1 o2) (a1, a2) (b1, b2)

B Proof of discr o = flatten · trie o (6.1)

Case o = OUnit:

flatten · trie OUnit
= { definition of trie }

flatten · TUnit · map val
= { definition of flatten }

return · map val
= { definition of discr }

discr OUnit

Case o = OSum o1 o2:

flatten · trie (OSum o1 o2)
= { definition of trie }

flatten · TSum · (trie o1 · filter (inl◦ × id) � . . .)
= { definition of flatten }

(flatten · outl ++ . . .) · (trie o1 · filter (inl◦ × id)� . . .)
= { fusion and computation }

flatten · trie o1 · filter (inl◦ × id) ++ . . .
= { ex hypothesi }

discr o1 · filter (inl◦ × id) ++ . . .
= { definition of discr }

discr (OSum o1 o2)

Case o = OProd o1 o2: here we make essential use of the fact that Trie K is a
functor and that flatten is natural in V.

flatten · trie (OProd o1 o2)
= { definition of trie }

flatten · TProd · Trie K1 (trie o2) · trie o1 · List curryl
= { definition of flatten }

concat · List flatten · flatten · Trie K1 (trie o2) · trie o1 · List curryl
= { flatten is natural: List f · flatten = flatten · Trie K f }

concat · List flatten · List (trie o2) · flatten · trie o1 · List curryl
= { ex hypothesi, twice }

concat · List (discr o2) · discr o1 · List curryl
= { definition of discr }

discr (OProd o1 o2) .

